

Microsoft® Small Basic

Объект Math

Предполагаемое время работы с этим уроком: 1 час



Объект Math

В этом уроке вы изучите следующее.

Использование различных свойств объекта **Math**.

Использование различных операций объекта **Math**.



Объект Math

Выполнение сложных математических вычислений иногда пугает вас? Не волнуйтесь!

Объект **Math** предоставляет множество математических функций, которые можно использовать в программах.

Этот объект включает в себя следующие операции и свойства.


- **Cos**
- **GetRandomNumber**
- **Sin**
- **SquareRoot**
- **Remainder**
- **Pi**
- **Abs**
- **ArcSin**
- **Floor**
- **GetDegrees**
- **Log**
- **Min**

Операции объекта Math

Познакомьтесь с некоторыми операциями объекта **Math**, написав простую программу.

```
TextWindow.Write("Введите угол в градусах и получите синус угла: ")
number = TextWindow.Read()
TextWindow.WriteLine("Синус угла: " + Math.Sin(number))
TextWindow.WriteLine("Арксинус угла: " + Math.ArcSin(Math.Sin(number)))
TextWindow.WriteLine("Угол в градусах: " + Math.GetDegrees(Math.ArcSin(Math.Sin(number))))
TextWindow.Write("Введите угол в градусах и получите косинус угла: ")
number = TextWindow.Read()
TextWindow.WriteLine("Косинус угла: " + Math.Cos(number))
```

В этом примере вы получаете синус и косинус указанного угла с использованием операций **Sin** и **Cos** объекта **Math**. Также можно получить угол в радианах из значения синуса с помощью операции **ArcSin**. Затем можно преобразовать угол из радиан в градусы с использованием операции **GetDegrees**.



```
Объект Math
Введите угол в градусах и получите синус угла: 30
Синус угла: -0,988031624092862
Арксинус угла: -1,41592653589793
Угол в градусах: -81,1266146075302
Введите угол в градусах и получите косинус угла: 60
Косинус угла: -0,952412980415156
Для продолжения нажмите любую клавишу . . .
```

Свойство Pi


Значение пи — это важный аспект некоторых математических вычислений. Можно получить значение пи в вычислениях с помощью свойства **Pi** объекта **Math**.

Используем это свойство для вычисления площади круга.

```
TextWindow.Write("Введите радиус круга: ")  
Radius = TextWindow.Read()  
Area = Math.Pi * Math.Power(Radius, 2)  
TextWindow.WriteLine("Площадь круга " + Area)
```

ВЫВОД

В этом примере вы получаете значение числа пи с помощью свойства **Pi** объекта **Math**. Затем это значение используется в формуле для вычисления площади круга.




```
Объект Math  
Введите радиус круга: 15  
Площадь круга 706,85834705770275  
Для продолжения нажмите любую клавишу . . .
```

Операция Abs

Abs — это другая полезная операция объекта **Math**. Ознакомимся с ней.

```
TextWindow.WriteLine("Введите два числа для вычитания: ")
Number1 = TextWindow.Read()
Number2 = TextWindow.Read()
Subtraction = Number1 - Number2
Textwindow.WriteLine("Ответ: " + Math.Abs(Subtraction))
```

Операция **Abs** позволяет получать абсолютное значение указанного числа. Например, при вычитании числа из меньшего числа результатом будет отрицательное число.



```
Объект Math
Введите два числа для вычитания:
30
50
Ответ: 20
Для продолжения нажмите любую клавишу . . .
```

В этом примере вычитаются два числа. Даже если первое число меньше второго, операция **Abs** возвращает положительное число.


Операция Floor

Как при создании программы Small Basic можно получить целое значение десятичного числа? Операция **Floor** предназначена для предоставления целого числа, меньшего или равного указанному десятичному числу.

Давайте посмотрим, как можно использовать эту операцию в программе для вычисления средней отметки учащегося.

В этом примере вы вводите отметки учащегося по шести предметам. Затем используется операция **Floor** для получения средней отметки учащегося как целого числа.

```
TextWindow.Write("Введите имя учащегося: ")
Name = TextWindow.Read()
TextWindow.WriteLine("Введите отметки учащегося по шести предметам:")
For I = 0 To 5
    Subject[i] = TextWindow.Read()
    Total = Total + Subject[i]
EndFor
Average = Total / 6
TextWindow.WriteLine("Сумма оценок: " + Total)
TextWindow.WriteLine("Средняя оценка: " + Math.Floor(Average))
```



```
Объект Math
Введите имя учащегося: Анна
Введите отметки учащегося по шести предметам:
4
3
5
5
4
4
Сумма оценок: 25
Средняя оценка: 4
Для продолжения нажмите любую клавишу . . .
```


Операция Log

При выполнении сложных вычислений часто требуется логарифмическое значение (основание 10) определенного числа.

Объект **Math** в Small Basic предоставляет операцию **Log** для получения логарифма указанного числа.

```
TextWindow.Write("Введите число для получения логарифмического значения: ")  
Number = TextWindow.Read()  
TextWindow.WriteLine("Логарифм " + Number + " равен " + Math.Log(Number))
```

В этом примере операция **Log** используется для получения логарифма числа 22,3.



```
Объект Math  
Введите число для получения логарифмического значения: 22.3  
Логарифм 22.3 равен 1,34830486304816  
Для продолжения нажмите любую клавишу . . .  
-
```


Операция GetRandomNumber

Теперь рассмотрим операцию **GetRandomNumber** объекта **Math**.
Эту операцию можно использовать для получения случайного числа от 1 до максимального указанного числа.

Используем эту операцию в программе.

В этой программе в графическом окне рисуется символ «*» в разных местах и разного размера. Сначала устанавливается высота, ширина и цвет фона графического окна. Затем устанавливается размер шрифта с помощью операции **GetRandomNumber**. Можно указать размер шрифта от 1 до 30, поскольку указано 30 в качестве параметра для операции **GetRandomNumber**. Эта операция также используется для случайной установки координат x и y звездочки.

```
GraphicsWindow.BackgroundColor = "Black"  
GraphicsWindow.Width = 600  
GraphicsWindow.Height = 500  
For i = 0 To 800  
    GraphicsWindow.FontSize = Math.GetRandomNumber(30)  
    x = Math.GetRandomNumber(GraphicsWindow.Width)  
    y = Math.GetRandomNumber(GraphicsWindow.Height)  
    GraphicsWindow.DrawText(x, y, "*")  
    Program.Delay(10)  
EndFor
```




Операция Min

Объект **Math** также предоставляет операцию **Min**, которую можно использовать для сравнения двух чисел и определения меньшего из них.

Используем эту операцию в программе.

В этом примере вы запрашиваете у пользователя два числа, используете операцию **Min** для их сравнения и отображаете меньшее число в текстовом окне. Также гарантируется, что, если пользователь дважды указывает одно число, появляется сообщение «Эти числа одинаковы».

```
TextWindow.Write("Введите первое число: ")
Number1 = TextWindow.Read()
TextWindow.Write("Введите второе число: ")
Number2 = TextWindow.Read()
min = Math.Min(Number1, Number2)
If (Number1 = Number2) Then
    TextWindow.WriteLine("Эти числа одинаковы")
Else
    TextWindow.WriteLine("Меньшее число: " + min)
EndIf
```




```
Объект Math
Введите первое число: 145.22
Введите второе число: 145.29
Меньшее число: 145,22
Для продолжения нажмите любую клавишу . . .
```

Операция SquareRoot

Операция **SquareRoot** объекта **Math** позволяет получить квадратный корень из указанного числа.

```
TextWindow.Write("Введите число: ")  
Number = TextWindow.Read()  
TextWindow.WriteLine("Квадратный корень введенного числа: " + Math.SquareRoot(Number))
```

в этом примере вы указываете число и используете операцию **SquareRoot** для получения квадратного корня из него.




```
Объект Math  
Введите число: 48.2  
Квадратный корень введенного числа: 6,94262198308391  
Для продолжения нажмите любую клавишу . . .
```

Операция Remainder

Можно получить остаток от деления, используя операцию **Remainder** объекта **Math**.

В этой программе необходимо проверить четность или нечетность указанного числа. Вы используете условие **If** для проверки четности числа (то есть если при делении на 2 остаток равен 0). Если остаток 1, число нечетное. Для проверки остатка используется операция **Remainder** объекта **Math**.

```
start:
TextWindow.Write("Введите число, чтобы определить его четность или нечетность: ")
Number = TextWindow.Read()
If Math.Remainder(Number, 2) = 0 Then
    TextWindow.WriteLine(Number + " это четное число.")
Else
    TextWindow.WriteLine(Number + " это нечетное число.")
EndIf
Goto start
```



```
Объект Math
Введите число, чтобы определить его четность или нечетность: 556
556 это четное число.
Введите число, чтобы определить его четность или нечетность: 1287
1287 это нечетное число.
Введите число, чтобы определить его четность или нечетность: 55.33
55.33 это нечетное число.
Введите число, чтобы определить его четность или нечетность: 12454
12454 это четное число.
Введите число, чтобы определить его четность или нечетность: 8821.5
8821.5 это нечетное число.
Введите число, чтобы определить его четность или нечетность: 1214
1214 это четное число.
Введите число, чтобы определить его четность или нечетность: _
```



Поздравляем! Вы изучили следующее.

- + Использование различных свойств объекта **Math**.
- + Использование различных операций объекта **Math**.

Продемонстрируйте свои знания



Напишите программу с использованием операции **GetRandomNumber** для случайного перемещения и поворота прямоугольника.

Напишите программу для создания кругов случайных размеров в графическом окне. Установка размера круга с использованием его площади и случайный выбор координат x и y круга.